
% ------------------------------------------------------------------------------------------------------- 
% soucet(+N1,+N2,-N)    N1,N2,N seznamy cifer třech čísel, 
%                       číslo N je součtem čísel N1 a N2 
%    V seznamech se mohou vyskytovat buď cifry nebo proměnné 
%    Za dvě různé proměnné musí být substituovány různé cifry 
% ------------------------------------------------------------------------------------------------------- 
 
soucet(N1,N2,N):- 
   sum1(N1,N2,N,0,0,[0,1,2,3,4,5,6,7,8,9],_). 
 
 
% ------------------------------------------------------------------------------------------------------- 
% sum1(N1,N2,N,P1,P,Cif1,Cif)    
%  N  je " skoro součtem"  N1 a N2, přičemž 
% P1 je přenos zprava (t.j. číslo, které se přičte k N1+N2),  
% P je přenos do leva (t.j. cifra, která by stála bude před N v součtu N1+N2), 
%  Cif1 je seznam cifer, které můžeme při řešení použít 
%  Cif je seznam, který vznikne tak, že ze seznamu Cif1 odstraníme 
%      cifry, které jsme při hledání řešení použili. 
% ------------------------------------------------------------------------------------------------------- 
 
sum1([],[],[],0,0,Cif,Cif). 
sum1([C1|N1],[C2|N2],[C|N],P1,P,Cif1,Cif):- 
   sum1(N1,N2,N,P1,P2,Cif1,Cif2), 
   soucet_cifer(C1,C2,C,P2,P,Cif2,Cif). 
 
 
% ------------------------------------------------------------------------------------------------------- 
% soucet_cifer(C1,C2,C,P1,P,Cif1,Cif) 
%    C1, C2, C jsou cifry, 
% P1 přenos zprava, P přenos doleva, 
% platí C1+C2+P1=10*P+C, 
% za C1, C2, C (pokud jsou to volné proměnné) můžeme použít cifry z Cif1, 
%        seznam Cif vznikne ze seznamu Cif1 vynecháním použitých cifer 
% ------------------------------------------------------------------------------------------------------- 
 
soucet_cifer(C1,C2,C,P1,P,Cif1,Cif):- 
   del(C1,Cif1,Cif2),             % vyber volnou cifru pro C1 
   del(C2,Cif2,Cif3),             % vyber volnou cifru pro C2 
   del(C,Cif3,Cif),               % vyber volnou cifru pro C 
   S is C1+C2+P1, 
   C is S mod 10, 
   P is S // 10. 
 
 



 
% ------------------------------------------------------------------------------------------------------- 
% del(X,+Z1,-Z2)  
%  Je-li X volná proměnná, substituuje za ní cifru vybranou ze Z1 a 
%     současně zkopíruje do Z2 všechny cifry 
%     mimo vybrané cifry 
%  Není-li X volná proměnná zkopíruje Z1 do Z2. 
% ------------------------------------------------------------------------ 
 
del(A,L,L):- 
   nonvar(A),!.       % A uz ma hodnotu 
del(A,[A|L],L). 
del(A,[B|L],[B|L1]):- 
   del(A,L,L1). 
 
%  nějaké algebrogramy 
 
algebrogram1([D,O,N,A,L,D], 
             [G,E,R,A,L,D], 
             [R,O,B,E,R,T]). 
 
algebrogram2([0,S,E,N,D], 
             [0,M,O,R,E], 
             [M,O,N,E,Y]). 
 
 
%  možná volání 
%    algebrogram1(N1,N2,N),soucet(N1,N2,N). 
%    algebrogram2(N1,N2,N),soucet(N1,N2,N), 
%           N1=[A,B|Z1],B>0                 % S>0 
%           N2=[C,D|Z2],D>0                 % M>0 
 


